
Part I: The �Prolog Language

� Types and Terms

� First-Order Horn Clauses

� Implication and Universal Quanti�cation in Goals

(First-Order Harrop Formulas)

� �-terms and Quanti�cation over Functions and Predicates

(Higher-Order Horn Clauses)

� �Prolog (Higher-Order Hereditary Harrop Formulas)

� The Module System

� The L

�

Sublanguage

1

Sublanguages of �Prolog

fohc

L

�

fohh

hohc

hohh

-

- -

fohc �rst-order Horn clauses

fohh �rst-order hereditary Harrop formulas

hohc higher-order Horn clauses

hohh higher-order hereditary Harrop formulas

Extensions to Prolog

� polymorphic typing

� Due to hohc:

� higher-order programming

� �-terms as data structures

� Due to fohh:

� modular programming

� abstract data types

� hypothetical reasoning

Kinds and Kind Declarations

� Primitive Types

� System Types

kind o type. (for propositions)

kind int type.

kind real type.

kind string type.

� User-De�ned Types, e:g:,

kind node type.

� Type Constructors

kind list type -> type.

kind pair type -> type -> type.

Types

� System types: o, int, real, string

� User-introduced primitive types: node

� Type variables (denoted by capital letters)

� Constructed types

list string, pair int (list string)

� Functional types (includes predicate types)

int -> real -> string

int -> int -> o

o -> int -> o

(int -> int) -> real

list A -> (A -> B) -> list B -> o

Note: ! associates to the right, e:g:,

�

1

! �

2

! �

3

denotes �

1

! (�

2

! �

3

).

Declarations and Terms

type :: A -> list A -> list A.

infixr :: 5.

type nil list A.

kind a,b,c type.

type f a -> b -> c.

type s a.

type t b.

Term Syntax

t ::= c j X j x\t j X\t j (t

1

t

2

)

Curried Notation is Used

((f s) t) or (f s t) instead of f(s,t).

(f s) also allowed.

Clauses and Goals

type true o. infixr , 2.

type , o -> o -> o. infixr ; 3.

type :-, =>, ; o -> o -> o. infixr :- 1.

type pi, sigma (A -> o) -> o. infixr => 4.

type append list A -> list A -> list A -> o.

append nil K K.

append (X :: L) K (X :: M) :- append L K M.

?- append (1 :: nil) (2 :: nil) L.

L == (1 :: 2 :: nil).

First-Order Horn Clauses

� Atomic Formulas:

A of type o whose top-level symbol is not a logical constant.

� Goal Formulas:

G ::= > j A j G

1

^G

2

j G

1

_G

2

j 9

�

x G

� De�nite Clauses:

D := A j G � A j 8

�

x D

Explicit Quanti�cation

(9x B

1

� B

2

) � 8x (B

1

� B

2

)

type adj, path node -> node -> o.

path X Y :- adj X Z, path Z Y.

path X Y :- sigma z\(adj X z, path z Y).

pi x\(pi y\(path x y :-

sigma z\(adj x z, path z y))).

?- path a X.

?- sigma x\(path a x).

Type and Clausal Order

� Order of a type expression:

ord(� ) = 0 (for atomic type or type variable � )

ord(�

1

! �

2

) = max(ord(�

1

) + 1; ord(�

2

))

� Clausal order:

ord(A) = 0 (if A is atomic or >)

ord(B

1

^B

2

) = max(ord(B

1

); ord(B

2

))

ord(B

1

_B

2

) = max(ord(B

1

); ord(B

2

))

ord(B

1

� B

2

) = max(ord(B

1

) + 1; ord(B

2

))

ord(8x B) = ord(B)

ord(9x B) = ord(B)

First-Order Restrictions

� Types in type declarations are of order 0 or 1 (no nesting of !

to the left). Also, o only occurs as a target type. Note that the

types of pi and sigma are exceptions.

Example:

int, int -> int, int -> o, int -> int -> int

But not:

(int -> int) -> int, o -> o

� Clausal order is either 0 or 1.

Example:

adj a b.

path X Y :- adj X Z, path Z Y.

11

First-Order Hereditary Harrop Formulas

� Goal Formulas:

G ::= > j A j G

1

^G

2

j G

1

_G

2

j 9

�

x G j D � G j 8

�

x G

� De�nite Clauses:

D := A j G � A j 8

�

x D

� First-order restrictions hold.

Goal-Directed Search

Goal-directed search is formalized with respect to uniform proofs.

See [Miller et. al., APAL 91]. Nondeterministic search is complete

with respect to intuitionistic provability.

Let � be a set of type declarations and let P be set of program

clauses. Six primitive operations describe goal-directed search.

AND To prove G

1

^G

2

from h�;Pi, attempt to prove both G

1

and

G

2

from h�;Pi.

OR To prove G

1

_ G

2

from h�;Pi, attempt to prove either G

1

or

G

2

from h�;Pi.

INSTANCE To prove 9

�

x G from h�;Pi, pick a term t of type �

and attempt to prove [t=x]G from h�;Pi.

AUGMENT To prove D � G from h�;Pi, attempt to prove G

from h�;P[fDgi. Note that D is removed after the interpreter

succeeds or fails to prove G. Thus, the program grows and

shrinks dynamically in a stack based manner.

GENERIC To prove 8

�

x G from h�;Pi, introduce a new constant

c of type � and attempt to prove [c=x]G from h� [ fcg;Pi.

14

BACKCHAIN To prove an atomic goal A from h�;Pi, the current

program P must be considered.

� If there is a universal instance of a program clause which is

equal to A, then we have a proof.

� If there is a program clause with a universal instance of the

form G � A then attempt to prove G from h�;Pi.

� If neither case holds then there is no proof of A from h�;Pi.

15

Logic Variables and Uni�cation

� In INSTANCE a logic variable is used instead of \guessing" a

term.

� InBACKCHAIN logic variables are used to obtain a universal

instance of the clause, and uni�cation is used to match the goal

with the head of the clause.

� Note that the AUGMENT operation may result in program

clauses containing logic variables.

� Because the constant c in GENERIC must be new, uni�cation

must be modi�ed so that it prevents the variables in the goal

and program from being instantiated with terms containing c.

Equality and Conversion

� �-conversion:

�x:s = �y:s[y=x] provided y does not occur free in s.

� �-conversion:

(�x:s)t = s[t=x]

� �-conversion:

�x:(sx) = s provided x does not occur free in s.

�Prolog implements �-conversion as its notion of equality. The fol-

lowing terms are equivalent.

x\(f x) y\(f y) (g\x\(g x) f) f

�Prolog programs cannot determine the name of a bound variable.

Substitution and Quanti�cation

p X :- pi y\(q X y).

� Substitute (f a) for X.

p (f a) :- pi y\(q (f a) y).

� Substitute (f y) for X.

p (f y) :- pi y\(q (f y) y).

� Variable capture must be avoided.

p (f y) :- pi z\(q (f y) z).

18

Implication and Universal Quanti�cation in

Goals

kind bug,jar type.

type j jar.

type sterile, heated jar -> o.

type dead, bug insect -> o.

type in insect -> jar -> o.

sterile J :- pi x\(bug x => in x J => dead x).

dead B :- heated j, in B j, bug B.

heated j.

?- sterile j.

h�;Pi ?- pi x\(bug x => in x j => dead x).

h� [ fgg;Pi ?- bug g => in g j => dead g.

h� [ fgg;P [ fbug g, in g jgi ?- dead g.

19

Logic Variables in Programs

type reverse list A -> list A -> o.

type rev list A -> list A -> list A -> o.

type rv list A -> list A -> o.

reverse L K :-

pi L\(rev nil L L) =>

pi X\(pi L\(pi K\(pi M\(rev (X::L) K M :- rev L K (X::M)))))

=> rev L K nil.

?- reverse (1::2::nil) K.

reverse L K :-

rv nil K =>

pi X\(pi L\(pi K\(rv (X::L) K :- rv L (X::K))))

=> rv L nil.

20



Abstract Data Types

type empty stack -> o.

type enter, remove int -> stack -> stack -> o.

?- pi emp\(pi stk\(

empty emp =>

pi S\(pi X\(enter X S (stk X S))) =>

pi S\(pi X\(remove X (stk X S) S)) =>

sigma S1\(sigma S2\(sigma S3\(sigma S4\(sigma S5\

(empty S1, enter 1 S1 S2, enter 2 S2 S3,

remove A S3 S4, remove B S4 S5)))))).

A == 2, B == 1.

?- pi emp\(pi stk\( ... =>

sigma U\(empty U, enter 1 U V))).

no.

The term (stk 1 emp) is formed as an instance of V, but the goal

fails because emp cannot escape its scope.

21

Higher-Order Horn Clauses

� Atomic Formulas:

A is a term of type o whose top-level symbol is not a logical

constant, and which does not contain any occurrences of �.

� Rigid Atomic Formulas:

A

r

is an atomic formula whose top-level symbol is also not a

variable.

� Goal Formulas:

G ::= > j A j G

1

^G

2

j G

1

_G

2

j 9

�

x G

� De�nite Clauses:

D := A

r

j G � A

r

j 8

�

x D

� No restrictions on order of types. Restrictions on clausal or-

der still hold. Terms instantiating x also cannot contain any

occurrences of �.

Examples of Higher-Order Programs

type mappred (A -> B -> o) -> list A -> list B -> o.

type forevery (A -> o) -> list A -> o.

type forsome (A -> o) -> list A -> o.

type sublist (A -> o) -> list A -> list A -> o.

mappred P nil nil.

mappred P (X :: L) (Y :: K) :- P X Y, mappred P L K.

forevery P nil.

forevery P (X :: L) :- P X, forevery P L.

forsome P (X :: L) :- P X ; forsome P L.

sublist P (X::L) (X::K) :- P X, sublist P L K.

sublist P (X::L) K :- sublist P L K.

sublist P nil nil.

23

The mappred Program

type mappred (A -> B -> o) -> list A -> list B -> o.

mappred P nil nil.

mappred P (X :: L) (Y :: K) :- P X Y, mappred P L K.

type age person -> int -> o.

age bob 23.

age sue 24.

age ned 23.

?- mappred age (ned::bob::sue::nil) L.

L == (23::23::24::nil).

?- mappred age L (23::23::24::nil).

L == (ned::bob::sue::nil);

L == (bob::ned::sue::nil).

?- mappred (x\y\(age y x)) (23::24::nil) K.

K == (bob::sue::nil);

K == (ned::sue::nil).

The sublist Program

type sublist (A -> o) -> list A -> list A -> o.

sublist P (X::L) (X::K) :- P X, sublist P L K.

sublist P (X::L) K :- sublist P L K.

sublist P nil nil.

type male, female person -> o.

male bob.

female sue.

male ned.

?- sublist male (ned::bob::sue::nil) L.

L == (ned::bob::nil);

L == (ned::nil);

L == (bob::nil);

no

25

The forsome Program

type forsome (A -> o) -> list A -> o.

forsome P (X :: L) :- P X ; forsome P L.

male bob.

female sue.

male ned.

?- forsome female (ned::bob::sue::nil) L.

yes.

26

The forevery Program

type forevery (A -> o) -> list A -> o.

forevery P nil.

forevery P (X :: L) :- P X, forevery P L.

age bob 23.

age sue 24.

age ned 23.

?- forevery (x\(sigma y\(age x y))) (ned::bob::sue::nil).

yes.

?- forevery (x\(age x A)) (ned::bob::sue::nil).

no.

?- forevery (x\(age x A)) (ned::bob::nil).

A == 23.

27

Computing with �-terms

type mapfun (A -> B) -> list A -> list B -> o.

type reducefun (A -> B -> B) -> list A -> B -> B -> o.

mapfun F nil nil.

mapfun F (X :: L) ((F X) :: K) :- mapfun F L K.

reducefun F nil Z Z.

reducefun F (H::T) Z (F H R) :- reducefun F T Z R.

28



The mapfun Program

type mapfun (A -> B) -> list A -> list B -> o.

mapfun F nil nil.

mapfun F (X :: L) ((F X) :: K) :- mapfun F L K.

type g i -> i -> i.

type a,b i.

?- mapfun (x\(g a x)) (a::b::nil) L.

L == ((g a a)::(g a b)::nil).

The interpreter forms the terms

((x\(g a x)) a) and ((x\(g a x)) b)

and reduces them.

?- mapfun F (a::b::nil) ((g a a)::(g a b)::nil).

F == x\(g a x);

no.

The interpreter tries the 4 uni�ers for (F a) and (g a a)

in the following order.

x\(g x x) x\(g a x) x\(g x a) x\(g a a)

29

Computing with �terms is not Functional

Programming

An alternative de�nition of mapfun illustrating that it is weaker

than mappred.

type mapfun (A -> B) -> list A -> list B -> o.

mapfun F L K :- mappred (x\y\(y = F x)) L K.

Computing with �terms involves uni�cation and conversion, but not

function computation. The following goal is not provable.

?- mapfun F (a::b::nil) (c::d::nil).

no.

The reducefun Program

type reducefun (A -> B -> B) -> list A -> B -> B -> o.

reducefun F nil Z Z.

reducefun F (H::T) Z (F H R) :- reducefun F T Z R.

?- reducefun (x\y\(x + y))) (3::4::8::nil) 6 R, S is R.

R == 3 + (4 + (8 + 6))

S == 21.

?- reducefun F (4::8::nil) 6 (1 + (4 + (1 + (8 + 6)))).

F == x\y\(1 + (4 + (1 + (8 + 6))));

F == x\y\(1 + (x + (1 + (8 + 6))));

F == x\y\(1 + (x + y));

no.

?- pi z\(reducefun F (4::8::nil) z (1 + (4 + (1 + (8 + z)))).

F == x\y\(1 + (x + y));

no.

Higher-Order Hereditary Harrop Formulas

� Goal Formulas:

G ::= > j A j G

1

^G

2

j G

1

_G

2

j 9

�

x G j D � G j 8

�

x G

� De�nite Clauses:

D := A j G � A j 8

�

x D

� No restrictions on order of types or on clausal order. The restric-

tion that atomic terms and substitution terms cannot contain

occurrences of � still holds.

� New restriction: the head of any atomic formula that appears in

a D formula cannot be a variable that is essentially existentially

quanti�ed.

Essentially Existential and Universal

Occurrences

� If a subformula occurs to the left of an even number of occur-

rences of � in a goal formula, then it is a positive subformula

occurrence. If it occurs to the left of an odd number of oc-

currences of �, it is a negative subformula occurrence. These

de�nitions are reversed for clauses.

� A bound variable occurrence is essentially universal if it is bound

by a positive occurrence of a universal quanti�er, by a negative

occurrence of an existential quanti�er, or by a �-abstraction.

Otherwise, it is essentially existential.

� In terms of the �Prolog interpreter, variables that get instanti-

ated with logic variables are essentially existential, while vari-

ables that get instantiated with new constants are essentially

universal.

Logical Foundation of �Prolog

� Based on Church's Simple Theory of Types [Church 40, JSL]

� The type o for formulas, and the quanti�ers pi and sigma

adopted directly.

� Di�erences

� Di�erent logical connectives are taken as primitive.

� Intuitionistic instead of classical logic is used.

� Type variables and constructors are allowed.

34

More Implementations of reverse

type reverse list A -> list A -> o.

reverse L K :- pi rev\(

pi L\(rev nil L L) =>

pi X\(pi L\(pi K\(pi M\(rev (X::L) K M :- rev L K (X::M)))))

=> rev L K nil).

reverse L K :- pi rv\(

rv nil K =>

pi X\(pi L\(pi K\(rv (X::L) K :- rv L (X::K))))

=> rv L nil).

35

Discharging a Constant from a Term

h�;Pi ?- pi y\(append (1::2::nil) y X).

h� [ fkg;Pi ?- append (1::2::nil) k X.

The term (1::2::k) is formed as an instance of X, but as seen

before, the goal fails because k cannot escape its scope.

h�;Pi ?- pi y\(append (1::2::nil) y (H y)).

h� [ fkg;Pi ?- append (1::2::nil) k (H k).

The terms (H k) and (1::2::k) are uni�ed. Of the two uni�ers,

w\(1::2::k) and w\(1::2::w), only the second is possible. It is

the result of discharging k from the term (1::2::k).

�Prolog's Module System

1. One-line header

module moduleName.

2. Preamble (4 directives)

accumulate, import, local, localkind

3. Declarations (which form the signature) and clauses

37

The accumulate directive

� Used to incorporate other modules as if they were typed at the

beginning of the current module.

� The signature of the module and of all of the modules named by

the accumulate directive must be successfully pairwise merged.

� Two signatures can be merged when:

� If a token has a kind declaration in both signatures, the dec-

larations must be identical.

� If a token has a type declaration in both signatures, the types

must be the same up to renaming of type variables.

� If a token has a type declaration in both signatures, if it also

has an in�x declaration in one signature, it must have the

same in�x declaration in the other.

Example Modules using accumulate

module mod1.

kind item type.

type p,q item -> o.

p X :- q X.

module mod2.

accumulate mod1.

type a item.

q a.

module mod3.

kind item type.

type p,q item -> o.

type a item.

p X :- q X.

q a.

Modules mod2 and mod3 have the same signature and program.

Declaring local scope to constants

� Universal quanti�cation in goals, e:g:, 8x(D � G), can be used

to introduce a new scoped constant. Note that this formula is

equivalent to (9x D) � G.

� Modules as existentially quanti�ed program clauses provides lo-

cal scoping:

E ::= D j 9

�

x E j E

1

^E

2

� No need to change the interpreter. A goal of the form E � G can

be expanded to one that doesn't contain existential quanti�ers

in clauses by using the equivalence (9x D) � G � 8x(D � G).

40



Example Module using local

module stack.

kind stack type -> type.

type empty stack A -> o.

type enter, remove A -> stack A -> stack A -> o.

local emp stack A.

local stk A -> stack A -> stack A.

empty emp.

enter X S (stk X S).

remove X (stk X S) S.

41

The import directive

module mod1.

import mod2 mod3.

� The clauses in mod2 and mod3 are available during the search

for proofs of the body of clauses in mod1. Logically...

� Suppose E

2

and E

3

are the formulas associated with mod2 and

mod3 and G � A is a clause in mod1.

� Then the clause used by the interpreter is really the one that is

equivalent to

((E

2

^E

3

) � G) � A

after existential quanti�ers in E

2

and E

3

are changed to univer-

sal quanti�ers over G.

42

Example Module using import

module int_stack.

import stack.

type stack_test int -> int -> o.

stack_test A B :-

sigma S1\(sigma S2\(sigma S3\(sigma S4\(sigma S5\

(empty S1, enter 1 S1 S2, enter 2 S2 S3,

remove A S3 S4, remove B S4 S5))))).

?- stack_test A B.

A == 2, B == 1.

The L

�

Sublanguage

� Restricts �Prolog by placing the following restriction on vari-

ables:

For every subterm in formula B of the form xy

1

: : : y

n

(n � 0) where x is essentially existentially quanti�ed in

B, the variables y

1

; : : : ; y

n

must be distinct variables that

are essentially universally quanti�ed within the scope of

the binding for x.

� Simpli�es �-conversion: all �-redexes have the form ty

1

: : : y

n

where we can assume that t has the form �y

1

: : : �y

n

:t

0

. By �-

reduction (�y

1

: : : �y

n

:t

0

)y

1

: : : y

n

simply reduces to t

0

.

� Simpli�es uni�cation: it is decidable and most general uni�ers

exist; it can be implemented with a simple extension to �rst-

order uni�cation.

L

�

Uni�cation Examples

� An example in L

�

x\y\(g (U x z) (V y)) = v\w\(X w)

U == x\y\(W y) X == w\(g (W w) (V w))

� An example that is not in L

�

(F a) = (g a a)

F == x\(g x x) F == x\(g a x)

F == x\(g x a) F == x\(g a a)

Part II: Specifying Logics and Inference Systems

� Specifying Syntax

� A Program for Computing Negation Normal Forms

� Example Speci�cations

� Natural Deduction

� A Sequent System

� A Modal Logic Speci�cation

� ��-Convertibility for the Untyped �-Calculus

� Evaluation for a Functional Language

� Correctness of Speci�cations

1

Interpreters for �Prolog

We distinguish between two kinds of interpreters for �Prolog.

� Speci�cations are with respect to a non-deterministic interpreter

(which is complete with respect to intuitionistic provability).

� The deterministic interpreter which provides an ordering on

clause and goal selection and uses a depth-�rst search discipline

with backtracking as in Prolog is used for actual implementa-

tions.

2

Why Theorem Proving as an Application?

� Speci�cation

� The declarative nature of programs allows natural speci�ca-

tions of a variety of logics as well as of the tasks involved in

theorem proving.

� �-terms are useful for expressing the higher-order abstract

syntax of object logics.

� Universal quanti�cation and implication in goal formulas are

useful for specifying various inference systems naturally and

directly.

� Implementation

� Search is fundamental to theorem proving.

� Uni�cation can be used to solve certain equations between

objects (e:g:, formulas, proofs).

� �-conversion can be used to implement substitution directly.

3



A First-Order Object Logic

kind form type.

kind i type.

type and, or, imp form -> form -> form.

type neg form -> form.

type forall (i -> form) -> form.

type exists (i -> form) -> form.

type false form.

type c i. infixl or 4.

type f i -> i -> i. infixl and 5.

type q form. infixr imp 6.

type p i -> form.

8x9y (p(x) � p(y))

(forall x\(exists y\((p x) imp (p y))))

4

Negation Normal Form

::A � A

:(A ^ B) � :A _ :B

:(A _ B) � :A ^ :B

:(A � B) � A ^ :B

:(8xA) � 9x(:A)

:(9xA) � 8x(:A)

5

Negation Normal Form Clauses I

kind nnf form -> form -> o.

nnf (A and B) (C and D) :- nnf A C, nnf B D.

nnf (A or B) (C or D) :- nnf A C, nnf B D.

nnf (A imp B) (C or D) :- nnf (neg A) C, nnf B D.

nnf (forall A) (forall B) :- pi x\(nnf (A x) (B x)).

nnf (exists A) (exists B) :- pi x\(nnf (A x) (B x)).

6

Negation Normal Form Clauses II

nnf (neg (neg A)) B :- nnf A B.

nnf (neg (A and B)) (C or D) :-

nnf (neg A) C, nnf (neg B) D.

nnf (neg (A or B)) (C and D) :-

nnf (neg A) C, nnf (neg B) D.

nnf (neg (A imp B)) (C and D) :-

nnf A C, nnf (neg B) D.

nnf (neg (forall A)) (exists B) :-

pi x\(nnf (neg (A x)) (B x)).

nnf (neg (exists A)) (forall B) :-

pi x\(nnf (neg (A x)) (B x)).

7



Natural Deduction I

A B

^-I

A ^ B

A

_-I

A _B

B

_-I

A _ B

(A)

B

�-I

A � B

(A)

?

:-I

:A

[t=x]A

9-I

9xA

[y=x]A

8-I

8xA

Proviso on 8-I: y cannot appear free in 8xA, or in any assumption

on which the deduction of [y=x]A depends.

8

Natural Deduction II

A ^ B

^-E

A

A ^B

^-E

B

A A � B

�-E

B

8xA

8-E

[t=x]A

A _B

(A)

C

(B)

C

_-E

C

A :A

:-E

?

9xA

([y=x]A)

B

9-E

B

?

?

I

A

(:A)

?

?

C

A

Proviso on 9-E rule: y cannot appear free in 9xA, in B, or in any

assumption on which the deduction of the upper occurrence of B

depends.

9

A Natural Deduction Proof

p(a) _ p(b)

p(a)

9-I

9x p(x)

p(b)

9-I

9x p(x)

_-E

9x p(x)

�-I

p(a) _ p(b) � 9x p(x)

10

Specifying Natural Deduction Rules

A B

^-I

A ^ B

A

_-I

A _ B

B

_-I

A _ B

type proof nprf -> form -> o.

type and_i nprf -> nprf -> nprf.

proof (and_i P1 P2) (A and B) :-

proof P1 A, proof P2 B.

proof (or_i P) (A or B) :-

proof P A; proof P B.

11



Specifying Existential Introduction

[t=x]A

9-I

9xA

type exists_i nprf -> nprf.

proof (exists_i P) (exists A) :- proof P (A T).

type exists_i i -> nprf -> nprf.

proof (exists_i T P) (exists A) :- proof P (A T).

12

Specifying Universal Introduction

[y=x]A

8-I

8xA

Proviso on 8-I: y cannot appear free in 8xA, or in any assumption

on which the deduction of [y=x]A depends.

proof (forall_i P) (forall A) :-

pi y\(proof (P y) (A y)).

type forall_i (i -> nprf) -> nprf.

13

Specifying the Discharge of Assumptions

(A)

B

�-I

A � B

proof (imp_i P) (A imp B) :-

pi pA\((proof pA A) => (proof (P pA) B)).

type imp_i (nprf -> nprf) -> nprf.

14

Example Execution

q

�-I

q � q

proof (imp_i P) (A imp B) :-

pi pA\((proof pA A) => (proof (P pA) B)).

h�;Pi ?- proof R (q imp q).

h�;Pi ?- pi pA\((proof pA q) => (proof (R1 pA) q)).

h� [ fpag;Pi ?- (proof pa q) => (proof (R1 pa) q)).

h� [ fpag;P [ fproof pa qg ?- proof (R1 pa) q.

Uni�cation Problem: R = (imp_i R1), (R1 pa) = pa

Solution: R1 := x\x, R := (imp_i x\x)

Not a Solution: R1 := x\pa

15



Elimination Rules

A ^B

^-E

A

A ^B

^-E

B

A A � B

�-E

B

8xA

8-E

[t=x]A

A _B

(A)

C

(B)

C

_-E

C

A :A

:-E

?

9xA

([y=x]A)

B

9-E

B

proof (and_e P) A :- proof P (A and B); proof P (B and A).

proof (forall_e P) (A T) :- proof P (forall A).

proof (exists_e P1 P2) B :- proof P1 (exists A),

pi y\(pi p\ ((proof p (A y)) => (proof (P2 Y p) B))).

proof (or_e P P1 P2) C :- proof P (A or B),

pi pA\((proof pA A) => (proof (P1 pA) C)),

pi pB\((proof pB B) => (proof (P2 pB) C)).

proof (imp_e P1 P2) B :- proof P1 A, proof P2 (A imp B).

proof (neg_e P1 P2) false :- proof P1 A, proof P2 (neg A).

16

Remaining Rules

(A)

?

:-I

:A

?

?

I

A

(:A)

?

?

C

A

proof (neg_i P) (neg A) :-

pi pA\((proof pA A) => (proof (P pA) false)).

proof (false_i P) A :- proof P false.

proof (false_c P) A :-

pi p\((proof p (neg A)) => (proof (P p) false)).

17

Sequent Calculus I

� �! A � �! B

^-R

� �! A ^ B

� �! A

_-R

� �! A _B

� �! B

_-R

� �! A _B

A;� �! B

�-R

� �! A � B

A;� �!?

:-R

� �! :A

� �! [y=x]A

8-R

� �! 8xA

� �! [t=x]A

9-R

� �! 9xA

Proviso on 8-R: y cannot appear free in the lower sequent.

18

Sequent Calculus II

A;B;� �! C

^-L

A ^B;� �! C

A;� �! C B;� �! C

_-L

A _B;� �! C

� �! A B;� �! C

�-L

A � B;� �! C

� �! A

:-L

:A;� �!?

[t=x]A;� �! C

8-L

8xA;� �! C

[y=x]A;� �! C

9-L

9xA;� �! C

� �!?

?-R

� �! A

A;� �! A (initial)

Proviso on 9-L: y cannot appear free in the lower sequent.

19



Specifying Sequent Systems

� �! A � �! B

^-R

� �! A ^ B

type --> (list form) -> form -> seq.

infix --> 4.

type proof sprf -> seq -> o.

type and_r sprf -> sprf -> sprf.

proof (and_r P1 P2) (Gamma --> (A and B)) :-

proof P1 (Gamma --> A), proof P2 (Gamma --> B).

20

Antecedent Rules

� �! A B;� �! C

�-L

A � B;� �! C

type memb A -> list A -> o.

memb X (X :: L).

memb X (Y :: L) :- memb X L.

type imp_l sprf -> sprf -> sprf.

proof (imp_l P1 P2) (Gamma --> C) :-

memb (A imp B) Gamma,

proof P1 (Gamma --> A),

proof P2 ((B::Gamma) --> C).

21

Initial Sequents

A;� �! A

proof (initial A) (Gamma --> A) :- memb A Gamma.

22

Classical Logic

� �! A;� � �! B;�

^-R

� �! A ^B;�

type --> (list form) -> (list form) -> seq.

proof (and_r P1 P2) (Gamma --> Delta) :-

memb (A and B) Delta,

proof P1 (Gamma --> (A::Delta)),

proof P2 (Gamma --> (B::Delta)).

23



Modal Sequents

R;� �! A

w

;� R;� �! B

w

;�

^-R

R;� �! (A ^ B)

w

;�

kind mform type.

kind world type.

kind wpair type.

type m form -> world -> mform.

type r world -> world -> wpair.

type mseq (list wpair) ->

(list form) -> (list form) -> seq.

proof (and_r P1 P2) (mseq R Gamma Delta) :-

memb (m (A and B) W) Delta,

proof P1 (mseq R Gamma ((m A W)::Delta)),

proof P2 (mseq R Gamma ((m B W)::Delta)).

24

The Modal Rules

[R ` r(w; x)] R;A

x

;� �!�

2-L

R; (2A)

w

;� �!�

R; r(w; x);� �!�; A

x

2-R

R;� �!�; (2A)

w

Proviso on 2-R: world x doesn't occur in the lower sequent.

[r(w; v); r(v; u) ` r(w; u)] r(w; v); r(v; u); p

u

�! p

u

2-L

r(w; v); r(v; u); (2

a

p)

w

�! p

u

2-R

r(w; v); (2

a

p)

w

�! (2

a

p)

v

2-R

(2

a

p)

w

�! (2

a

2

a

p)

w

�-R

�! (2

a

p � 2

a

2

a

p)

w

25

Specifying the Modal Introduction Rule

R; r(w; x);� �!�; A

x

2-R

R;� �!�; (2A)

w

type box form -> form.

proof (box_r P) (mseq R Gamma Delta) :-

memb (mform (box A) W) Delta,

pi x\(proof (P x)

(mseq ((r W x)::R) Gamma ((mform A x)::Delta))).

26

Specifying the Modal Elimination Rule

[R ` r(w; x)] R;A

x

;� �!�

2-L

R; (2A)

w

;� �!�

type related (list wpair) -> wpair -> o.

related R (r W X) :- memb (r W X) R.

related R (r W W).

related R (r W X) :- related R (r W Z), related R (r Z X).

proof (box_l P) (mseq R Gamma Delta) :-

memb (mform (box A) W) Gamma,

proof P (mseq R (mform A X) Gamma Delta),

related R (r W X).

27



��-Convertibility for the Untyped �-Calculus

kind tm type.

type app tm -> tm -> tm.

type abs (tm -> tm) -> tm.

�f�n:f(fn) �x:xx

(abs f\(abs n\(app f (app f n))))

(abs x\(app x x))

28

��-Redexes

� �-conversion: (�x:s)t = s[t=x]

� �-conversion: �x:(sx) = s provided x does not occur free in s.

type redex tm -> tm -> o.

redex (app (abs S) T) (S T).

redex (abs x\(app S x)) S.

29

One-Step Reducibility

M ! P

MN ! PN

N ! P

MN !MP

M ! N

�x:M ! �x:N

type red1 tm -> tm -> o.

red1 M N :- redex M N.

red1 (app M N) (app P N) :- red1 M P.

red1 (app M N) (app M P) :- red1 N P.

red1 (abs M) (abs N) :- pi x\(red1 (M x) (N x)).

30

��-Convertibility and Normalization

conv M N :- red1 M N.

conv M M.

conv M N :- conv N M.

conv M N :- conv M P, conv P N.

norm M N :- red1 M P, !, norm P N.

norm M M.

31



Correctness of Representation of �-terms

� An Encoding of Untyped Terms to Meta-Terms

� Given �: a mapping from the constants of the object lan-

guage to a �xed set of constants of the meta-language of

type tm.

� Given �: a mapping from the variables of the object language

to the meta-variables of type tm.

� Example:

��

�f�n:f(fn)

��

�
�

�

(abs f\(abs n\(app f (app f n))))

Theorem (Correctness of Encoding of Untyped Terms)

The encoding

�� ��

�
�

is a bijection from the �-equivalence classes of

untyped terms to the ��-equivalence classes of meta-terms of type

tm.

32

Correctness of ��-Convertibility Speci�cation

Theorem Let M and N be untyped terms. Then M =

��

N if and

only if

(conv

��

M

��

�
�

��

N

��

�
�

)

is provable.

33

Evaluation for a Functional Language

app : tm! tm! tm nil : tm

abs : (tm! tm)! tm cons : tm! tm! tm

0 : tm hd : tm! tm

s : tm! tm tl : tm! tm

true : tm empty : tm! tm

false : tm fix : (tm! tm)! tm

if : tm! tm! tm! tm let : (tm! tm)! tm! tm

app (abs M) N ! MN empty nil ! true

if true M N ! M empty (cons M N) ! false

if false M N ! N fix M ! M (fix M)

hd (cons M N) ! M let M N ! MN

tl (cons M N) ! N

34

A Speci�cation of Evaluation

type eval tm -> tm -> o.

eval (abs M N) P :- eval N N', eval (M N') P.

eval (if C M N) M' :- eval C tru, eval M M'.

eval (if C M N) N' :- eval C fals, eval N N'.

eval (hd L) M' :- eval L (cons M N), eval M M'.

eval (tl L) N' :- eval L (cons M N), eval N N'.

eval (empty L) tru :- eval L nill.

eval (empty L) fals :- eval L (cons M N).

eval (fix M) N :- eval (M (fix M)) N.

eval (let M N) P :- eval N N', eval (M N') P.

35



Some Related Languages

� The Logical Framework (LF) [Harper, Honsell, & Plotkin, JACM

93] is a type theory developed to capture the generalities across

a wide variety of object logics. A speci�cation of a logic in LF

can be \compiled" rather directly into a set of �Prolog clauses.

� The Forum logic programming language [Miller, TCS 96] imple-

ments an extension of higher-order hereditary Harrop formulas

(hohh) to linear logic.

� Isabelle [Paulson 94] is a \generic" tactic theorem prover imple-

mented in ML. It contains a speci�cation language which is a

subset of hohh. The two are very close in speci�cation strength.

36

Part III: Implementing Automatic Theorem

Provers

An Automatic Prover for First-Order Classical Logic

� A strategy for �nding sequent proofs

� An implementation using three subprocedures

1

Reversibility of Rules

A;B;� �!�

^-L

A ^ B;� �!�

� �! A;� � �! B;�

^-R

� �! A ^B;�

^-L: There is a proof of one of the formulas in � from A ^ B and

� if and only if there is a proof of one of the formulas in � from A

and B and �.

^-R: There is a proof of A ^B or of one of the formulas in � from

� if and only if there is a proof of A or one of the formulas in �

from � and there is a proof of B or one of the formulas in � from

�.

2

Non-Reversibility of Rules

The only two rules in the classical sequent calculus presented that

are not reversible are:

[t=x]A;� �!�

8-L

8xA;� �!�

� �! [t=x]A;�

9-R

� �! 9xA;�

For example, there may be a proof of one of the formulas in � from

8xA and �, but no term t such that there is a proof of one of the

formulas in � from [t=x]A and �. It may be the case that 8xA must

be instantiated with more than one term.

3



A Speci�cation that Removes Formulas

A;B;� �!�

^-L

A ^ B;� �!�

type memb_and_rest A -> (list A) -> (list A) -> o.

memb_and_rest A (A::L) L.

memb_and_rest A (B::L) (B::K) :- memb_and_rest A L K.

proof1 (and_l P) (Gamma --> Delta) :-

memb_and_rest (A and B) Gamma Gamma1,

proof1 P ((A::B::Gamma1) --> Delta).

4

Step 1 of 3: the proof1 procedure

1. Apply all rules except 8-L and 9-R until nothing more can be

done. The result is a set of sequents with atomic and universally

quanti�ed formulas on the left, and atomic and existentially quan-

ti�ed formulas on the right.

proof1 (initial A) (Gamma --> Delta) :-

memb A Gamma, memb A Delta.

proof1 (and_r P1 P2) (Gamma --> Delta) :-

memb_and_rest (A and B) Delta Delta1,

proof1 P1 (Gamma --> (A::Delta1)),

proof1 P2 (Gamma --> (B::Delta1)).

proof1 (imp_l P1 P2) (Gamma --> Delta) :-

memb_and_rest (A imp B) Gamma Gamma1,

proof1 P1 (Gamma1 --> (A::Delta)),

proof1 P2 ((B::Gamma1) --> Delta).

.
.
.

5

Step 2 of 3: the proof2 procedure

2. Apply all rules including versions of the rules for 8-L and 9-R

that remove the quanti�ed formula after applying the rule, and try

to complete the proof. Stop if a proof is successfully completed.

proof2 (forall_l P) (Gamma --> Delta) :-

memb_and_rest (forall A) Gamma Gamma1,

proof2 P (((A T)::Gamma1) --> Delta).

proof2 (exists_r P) (Gamma --> Delta) :-

memb_and_rest (exists A) Delta Delta1,

proof2 P (Gamma1 --> ((A T)::Delta1)).

.
.
.

(plus duplicates of each of the proof1 clauses)

6

Step 3 of 3: the nprove procedure

3. Add an additional copy of each quanti�ed formula to the sequents

obtained from step 1, and repeat steps 2 and 3.

nprove N P Seq :- amplify N Seq ASeq, proof2 P ASeq.

nprove N P Seq :- M is (N + 1), nprove M P Seq.

amplify 1 Seq Seq.

amplify N (Gamma1 --> Delta1) (Gamma2 --> Delta2) :-

N > 1,

amplify_forall N Gamma1 Gamma2,

amplify_exists N Delta1 Delta2.

7



add_copies 1 A L (A::L).

add_copies N A L (A::K) :-

N > 1, M is (N - 1),

add_copies M A L K.

amplify_forall N nil nil.

amplify_forall N ((forall A)::Gamma) Gamma2 :-

amplify_forall N Gamma Gamma1,

add_copies N (forall A) Gamma1 Gamma2.

amplify_forall N (A::Gamma) (A::Gamma1) :-

amplify_forall N Gamma Gamma1.

amplify_exists N nil nil.

amplify_exists N ((exists A)::Delta) Delta2 :-

amplify_exists N Delta Delta1,

add_copies N (exists A) Delta1 Delta2.

amplify_exists N (A::Delta) (A::Delta1) :-

amplify_exists N Delta Delta1.

8

Putting it all Together

The top-level predicate is proof1. Add one more clause for it at

the end.

proof1 P Seq :- nprove 1 P Seq.

nprove N P Seq :- amplify N Seq ASeq, proof2 P ASeq.

nprove N P Seq :- M is (N + 1), nprove M P Seq.

Completeness follows from the fact proved in [Andrews, JACM 81]

that duplication of outermost quanti�ers is all that is necessary to

obtain a complete procedure, and the fact that step 2 will always

terminate.

9

Examples

The �rst proof completes at ampli�cation 1. The second needs

ampli�cation 2.

p(a) �! p(a)

9-R

p(a) �! 9x p(x)

p(b) �! p(b)

9-R

p(b) �! 9x p(x)

_-L

p(a) _ p(b) �! 9x p(x)

�-R

�! p(a) _ p(b) � 9x p(x)

�! p(t) � p(z); p(z) � p(w)

8-R

�! p(t) � p(z); 8y (p(z) � p(y))

9-R

�! p(t) � p(z); 9x 8y (p(x) � p(y))

8-R

�! 8y (p(t) � p(y)); 9x 8y (p(x) � p(y))

9-R

�! 9x 8y (p(x) � p(y)); 9x 8y (p(x) � p(y))

amplify 2

�! 9x 8y (p(x) � p(y))

Part IV: Implementing Interactive Tactic

Theorem Provers

� Inference Rules as Tactics

� A Goal Reduction Tactical

� Some Common Tacticals

� Tactics and Tacticals for Interaction

� An Example Execution

1



Tactic Theorem Provers

� In general, more exibility in control of search is needed than

can be provided by depth-�rst search with backtracking.

� Tactics and tacticals have proven to be a powerful mechanism

for implementing theorem provers. Example tactic provers (all

ML implementations) include:

� LCF [Gordon, Milner, & Wadsworth]

� HOL [Gordon]

� Isabelle [Paulson]

� Nuprl [Constable et. al.]

� Coq [Huet et. al.]

� Tactics and tacticals can be implemented directly and naturally

in �Prolog. They implement an interpreter for goal-directed

theorem proving in the logic programming setting.

2

Inference Rules As Tactics

A B

^-I

A ^ B

proof (and_i P1 P2) (A and B) :-

proof P1 A, proof P2 B.

and_i_tac (proof (and_i P1 P2) (A and B))

((proof P1 A) ^^ (proof P2 B)).

type and_i_tac goal -> goal -> o.

type proof nprf -> form -> goal.

type ^^ goal -> goal -> goal.

infix ^^ 3.

3

Tactics with Assumption Lists

and_i_tac (proof (and_i P1 P2) (A and B))

((proof P1 A) ^^ (proof P2 B)).

kind judg type.

type proof nprf -> form -> judg.

type deduct (list judg) -> judg -> goal.

and_i_tac (deduct Gamma (proof (and_i P1 P2) (A and B)))

((deduct Gamma (proof P1 A)) ^^

(deduct Gamma (proof P2 B))).

4

Goal Constructors

type tt goal.

type ^^ goal -> goal -> goal.

type vv goal -> goal -> goal.

type all (A -> goal) -> goal.

type some (A -> goal) -> goal.

type ==>> o -> goal -> goal.

infixl ^^ 3.

infixl vv 3.

infixr ==>> 3.

5



A Goal Reduction Tactical

type maptac (goal -> goal -> o) -> goal -> goal -> o.

maptac Tac tt tt.

maptac Tac (InGoal1 ^^ InGoal2) (OutGoal1 ^^ OutGoal2) :-

maptac Tac InGoal1 OutGoal1,

maptac Tac InGoal2 OutGoal2.

maptac Tac (all InGoal) (all OutGoal) :-

pi x\(maptac Tac (InGoal x) (OutGoal x)).

maptac Tac (InGoal1 vv InGoal2) OutGoal :-

maptac Tac InGoal1 OutGoal; maptac Tac InGoal2 OutGoal.

maptac Tac (some InGoal) OutGoal :-

sigma T\(maptac Tac (InGoal T) OutGoal).

maptac Tac (D ==>> InGoal) (D ==>> OutGoal) :-

D => (maptac Tac InGoal OutGoal).

maptac Tac InGoal OutGoal :- Tac InGoal OutGoal.

6

Tacticals

then Tac1 Tac2 InGoal OutGoal :-

Tac1 InGoal MidGoal,

maptac Tac2 MidGoal OutGoal.

orelse Tac1 Tac2 InGoal OutGoal :-

Tac1 InGoal OutGoal; Tac2 InGoal OutGoal.

idtac Goal Goal.

repeat Tac InGoal OutGoal :-

orelse (then Tac (repeat Tac)) idtac InGoal OutGoal.

try Tac InGoal OutGoal :-

orelse Tac idtac InGoal OutGoal.

7

Interactive Theorem Proving

type query (goal -> o) -> goal -> goal -> o.

type inter (goal -> o) -> goal -> goal -> o.

type with_tacs string -> (goal -> goal -> o)

-> goal -> goal -> o.

query PrintPred InGoal OutGoal :-

PrintPred InGoal,

print "Enter tactic:", readtac Tac,

(Tac = backup, !, fail; Tac InGoal OutGoal).

inter PrintPred InGoal OutGoal :-

repeat (query PrintPred) InGoal OutGoal.

with_tacs M Tac InGoal OutGoal :-

M ==> (Tac InGoal OutGoal).

8

Interactive Tactics for Natural Deduction

� Allowing the User to Specify Substitution Terms

exists_i_tac (proof (exists_i P) (exists A))

(proof P (A T)).

exists_i_sub (proof (exists_i P) (exists A))

(proof P (A T)) :-

print "Enter substitution term:", read T.

� Adding Lemmas

modus_ponens (proof P A)

((proof Q B) ^^

((assump Q B) ==>> (proof P A))) :-

print "Enter lemma:", read B.

close_tac (proof P A) tt :- assump P A.

9



Natural Deduction Inference Rule Tactics

(A)

B

�-I

A � B

proof (imp_i P) (A imp B) :-

pi pA\((proof pA A) => (proof (P pA) B)).

imp_i_tac (proof (imp_i P) (A imp B))

(all pA\((assump pA A) ==>> (proof (P pA) B))).

imp_i_tac (deduct Gamma (proof (imp_i P) (A imp B)))

(all pA\(deduct ((proof pA A)::Gamma)

(proof (P pA) B)).

All introduction rules can be translated to tactics similarly.

10

Elimination Rules as Tactics

A ^B

^-E

A

A ^ B

^-E

B

proof (and_e P) A :-

proof P (A and B); proof P (B and A).

and_e_tac (deduct Gamma (proof P A))

((deduct Gamma (proof (and_e1 P) (A and B))) vv

(deduct Gamma (proof (and_e1 P) (B and A)))).

and_e_tac (deduct Gamma (proof P A))

((deduct Gamma (proof (and_e1 P) (A and B))) vv

(deduct Gamma (proof (and_e1 P) (B and A)))) :-

print "Enter second conjunct:", read B.

11

Forward Proof Using Elimination Rules

and_e_tac N (deduct Gamma (proof PC C))

(deduct ((proof (and_e1 P) A)::

(proof (and_e2 P) B)::Gamma)

(proof PC C)) :-

nth_item N (proof P (A and B)) Gamma.

All elimination rules can be implemented as tactics similarly.

12

An Example Query

?- interactive

(proof P (((q a) or (q b)) imp (exists x\(q x))))

OutGoal.

Assumptions:

Conclusion:

(q a or q b) imp (exists x\(q x))

Enter tactic: ?- imp_i_tac.

Assumptions:

1 q a or q b

Conclusion:

exists x\(q x)

Enter tactic: ?- exists_i_tac.

13



Assumptions:

1 q a or q b

Conclusion:

q T

Enter tactic: ?- or_e_tac 1.

Assumptions:

1 q a

2 q a or q b

Conclusion:

q T

Enter tactic: ?- close_tac 1.

14

Assumptions:

1 q b

2 q a or q b

Conclusion:

q a

Enter tactic: ?- backup.

.
.
.

Enter tactic: ?- backup.

.
.
.

Enter tactic: ?- backup.

15

Assumptions:

1 q a or q b

Conclusion:

exists x\(q x)

Enter tactic: ?- then (or_e_tac 1)

(then exists_i_tac

(close_tac 1)).

P = imp_i p\(or_e p p1\(exists_i a p1)

p2\(exists_i b p2))

OutGoal = all p\((all p1\tt) ^^ (all p2\tt))

16

Generic Theorem Proving

� Logics in the Isabelle theorem prover [Paulson 94] are speci�ed

in a language which is a subset of hohh, while control, including

tactics and tacticals, is implemented in ML.

� Here, tactics and tacticals are speci�ed in hohh. The �Prolog

interpreter associates control primitives (search operations) to

the logical connectives of hohh.

� Much work has gone into making Isabelle e�cient as well as

providing extensive environments for several particular object

logics. These environments include e�cient specialized tactics

as well as large libraries of theorems.

� Such an e�ort has not been made for �Prolog, but could be. Ex-

perience with Isabelle demonstrates the e�ectiveness of generic

theorem proving.

17



Part V: An Implementation of Higher-Order

Term Rewriting

� Higher-Order Rewrite Rules

� Some Example Rewrite Systems

� Expressing a Rewrite System as a Set of Tactics

� Tactics and Tacticals for Rewriting

1

Higher-Order Rewriting

� Higher-order rewrite systems use �-terms as a meta-language

for expressing the equality relation for object languages that

include notions of bound variables [Nipkow LICS'91, Klop 80,

Aczel 78]

� Many operations on formulas and programs can be naturally

expressed as higher-order rewrite systems.

� Capabilities for rewriting can be added to tactic style theorem

provers, used to reason about the equality relation of a par-

ticular object logic, and combined with other theorem proving

techniques.

� Higher-order logic programming allows:

� a natural speci�cation of higher-order rewrite systems

� powerful mechanisms for descending through terms and match-

ing terms with rewrite templates

2

Higher-Order Rewrite Rules

A rewrite rule is a pair l �! r such that l and r are simply-typed

�-terms of the same primitive type, l is a term in L

�

, and all free

variables in r also occur in l.

Example 1: ��-conversion for �-terms

� �-conversion: (�x:s)t = s[t=x]

� �-conversion: �x:(sx) = s provided x does not occur free in s.

type app tm -> tm -> tm.

type abs (tm -> tm) -> tm.

type redex tm -> tm -> o.

redex (app (abs S) T) (S T).

redex (abs x\(app S x)) S.

3

Three Parts of a Rewriting Procedure

� Rewrite Rules

redex (app (abs S) T) (S T).

redex (abs x\(app S x)) S.

� Congruence and One-Step Rewriting

red1 M N :- redex M N.

red1 (app M N) (app P N) :- red1 M P.

red1 (app M N) (app M P) :- red1 N P.

red1 (abs M) (abs N) :- pi x\(red1 (M x) (N x)).

� Multiple Step Reduction

reduce M N :- red1 M P, reduce P N.

reduce M M.

4



Rewriting in a Tactic Theorem Prover

� The previous example implements the leftmost-outermost rewrite

stategy. Using a di�erent order on the red1 clauses can give

other rewrite strategies such as bottom-up.

� In a tactic theorem prover, rewrite rules and congruence rules

can be implemented as basic tactics. More complex tactics can

be implemented for various strategies.

5

Rewrite and Congruence Rules as Tactics

type == A -> A -> goal.

infix == 7.

type prim goal -> goal.

type rew goal -> goal -> o.

type cong goal -> goal -> o.

type cong_const goal -> goal -> o.

rew (prim ((app (abs S) T) == (S T))) tt.

rew (prim ((abs x\(app S x)) == S)) tt.

cong (prim ((app M N) == (app P Q)))

((prim (M == P)) ^^ (prim (N == Q))).

cong (prim ((abs M) == (abs N)))

(all x\((cong_const (prim (x == x)) tt) ==>>

(prim ((M x) == (N x))))).

cong_const (prim (f == f)) tt.

6

Example 2: Evaluation as Rewriting

app : tm! tm! tm nil : tm

abs : (tm! tm)! tm cons : tm! tm! tm

0 : tm hd : tm! tm

s : tm! tm tl : tm! tm

true : tm empty : tm! tm

false : tm fix : (tm! tm)! tm

if : tm! tm! tm! tm let : (tm! tm)! tm! tm

7

Congruence Tactics for Evaluation

cong_const (prim (tru == tru)) tt.

cong_const (prim (fals == fals)) tt.

cong_const (prim (z == z)) tt.

cong_const (prim (nill == nill)) tt.

cong (prim ((s M) == (s N))) (prim (M == N)).

cong (prim ((cons M N) == (cons P Q)))

((prim (M == P)) ^^ (prim (N == Q))).

cong (prim ((hd M) == (hd N))) (prim (M == N)).

cong (prim ((tl M) == (tl N))) (prim (M == N)).

cong (prim ((empty M) == (empty N))) (prim (M == N)).

cong (prim ((if C M N) == (if D P Q)))

((prim (C == D)) ^^ (prim (M == P)) ^^ (prim (N == Q))).

cong (prim ((fix M) == (fix N)))

(all x\((cong_const (prim (x == x)) tt) ==>> (prim ((M x) == (N x))))).

cong (prim ((let M N) == (let P Q)))

((all x\((cong_const (prim (x == x)) tt) ==>>

(prim ((M x) == (P x))))) ^^ (prim (N == Q))).

8



Evaluation Rewrite Rules

app (abs M) N ! MN empty nil ! true

if true M N ! M empty (cons M N) ! false

if false M N ! N fix M ! M (fix M)

hd (cons M N) ! M let M N ! MN

tl (cons M N) ! N

9

Tactics Implementing Evaluation Rewrites

rew (prim ((app (abs M) N) == (M N))) tt.

rew (prim ((abs X (app M X)) == M)) tt.

rew (prim ((hd (cons M N)) == M)) tt.

rew (prim ((tl (cons M N)) == N)) tt.

rew (prim ((empty nill) == tru)) tt.

rew (prim ((empty (cons M N)) == fals)) tt.

rew (prim ((if tru M N) == M)) tt.

rew (prim ((if fals M N) == N)) tt.

rew (prim ((fix M) == (M (fix M)))) tt.

rew (prim ((let M N) == (M N))) tt.

10

Example 3: Negation Normal Forms

� Congruence Rules

cong (prim ((A and B) == (C and D)))

((prim (A == C)) ^^ (prim (B == D))).

cong (prim ((forall A) == (forall B)))

(all x\((cong_const (prim (x == x)) tt) ==>>

(prim ((A x) == (B x))))).

� Rewrite Rules

rew (prim ((neg (A and B)) ==

((neg A) or (neg B)))) tt.

rew (prim ((neg (forall A)) ==

(exists x\(neg (A x))))) tt.

11

A Modi�ed maptac

type maptacC (goal -> goal -> o) -> goal -> goal -> o.

maptacC Tac tt tt.

maptacC Tac (InGoal1 ^^ InGoal2) OutGoal :-

Tac (InGoal1 ^^ InGoal2) OutGoal.

maptacC Tac (all InGoal) (all OutGoal) :-

pi x\(maptacC Tac (InGoal x) (OutGoal x)).

maptacC Tac (InGoal1 vv InGoal2) OutGoal :-

maptacC Tac InGoal1 OutGoal;

maptacC Tac InGoal2 OutGoal.

maptacC Tac (some InGoal) OutGoal :-

sigma T\(maptacC Tac (InGoal T) OutGoal).

maptacC Tac (D ==>> InGoal) (D ==>> OutGoal) :-

D => (maptacC Tac InGoal OutGoal).

maptacC Tac (prim InGoal) OutGoal :-

Tac (prim InGoal) OutGoal.

12



Modi�ed Tacticals Using maptacC

thenC Tac1 Tac2 InGoal OutGoal :-

Tac1 InGoal MidGoal,

maptacC Tac2 MidGoal OutGoal.

repeatC Tac InGoal OutGoal :-

orelse (thenC Tac (repeatC Tac)) idtac InGoal OutGoal.

13

Rewrite Tactics and Tacticals I

refl (prim (M == N)) tt.

sym (prim (M == N)) (prim (N == M)).

trans (prim (M == N)) ((prim (M == P)) ^^ (prim (P == N))).

left Tac (prim InG) OutG :- Tac (prim InG) OutG.

left Tac (InG ^^ G) (OutG ^^ G) :-

maptacC (left Tac) InG OutG.

left_rew Tac InG OutG :-

thenC trans (left Tac) InG OutG.

14

Rewrite Tactics and Tacticals II

right Tac (prim InG) OutG :- Tac (prim InG) OutG.

right Tac (G ^^ InG) (G ^^ OutG) :-

maptacC (right Tac) InG OutG.

right_rew Tac InG OutG :-

thenC trans (right Tac) InG OutG.

first Tac (prim InG) OutG :- Tac (prim InG) OutG.

first Tac (InG ^^ G) (OutG ^^ G) :-

maptacC (first Tac) InG OutG, !.

first Tac (G ^^ InG) (G ^^ OutG) :-

maptacC (first Tac) InG OutG, !.

15

Bottom-Up Rewriting

bu Cong Rew InG OutG :-

then (bu_sub Cong Rew)

(orelse (then (left_rew Rew) (bu Cong Rew))

refl) InG OutG.

bu_sub Cong Rew InG OutG :-

try (left_rew (then Cong (bu Cong Rew))) InG OutG.

16



Leftmost-Outermost Rewriting

lo Cong Rew InG OutG :-

then (repeat (left_rew (lo_rew Cong Rew)))

refl InG OutG.

lo_rew Cong Rew InG OutG :-

orelse Rew (then (thenC Cong

(first (lo_rew Cong Rew)))

refl) InG OutG.

17

An Example

Let APP be the following term representing the program for ap-

pending two lists in our functional language.

(fix �F:(abs �l

1

:(abs �l

2

:

(if (empty l

1

) l

2

(cons (hd l

1

) (app (app F (tl l

1

)) l

2

))))))

� The lo strategy reduces

(app (app APP (cons 0 nil )) (cons (s 0) nil ))

to (cons 0 (cons (s 0) nil )):

The lo strategy corresponds to lazy evaluation of this language.

� The bu strategy loops, repeatedly applying the rewrite rule for

fix and expanding the de�nition of the function.

18

Other Rewrite Strategies

� The bu and lo tactics implement common complete strategies

for terminating rewrite systems. They illustrate the use of tac-

tics and tacticals for implementing rewrite procedures.

� The real power of the tactic setting is that it provides a set of

high-level primitives with which to write specialized strategies.

Examples include:

� Call-by-value vs. call-by-name evaluation. Strong vs. weak

evaluation (reducing under a �-abstraction or not).

[Hannan, ELP'91]

� Type-driven rewriting using �-expansion. [Pfenning,91]

� Layered rewriting where the application of a subset of the

possible rewrite rules are applied, and rewriting is interleaved

with other reasoning.

� Tactics specialized to particular applications or domains.

19

Part VI: Encoding the Logical Framework in

�Prolog

� Syntax of the Logical Framework (LF)

[Harper, Honsell, & Plotkin, JACM 93]

� An Example LF Signature

� Translating LF Signatures to Logic Programming Speci�cations

1



fohc

L

�

fohh

hohc

hohh

LF

�

�

�

�

�

�

�

�

�

�

�

�

�1

-

- -

6 6

� LF allows function types of any order, but does not allow Type

(which is the LF equivalent of o) anywhere in types except as a

target type. There is no quanti�cation over Type.

� Unlike L

�

which restricts the form of terms, LF extends them

to allow dependent types.

� Note that our example speci�cations don't use predicate quan-

ti�cation (though the implementation of tactics and tacticals

use it extensively). Our encoding \compiles" LF signatures into

the sublanguage of hohh without predicate quanti�cation.

2

LF Syntax

Syntax for LF Kinds, Types, Objects

K := Type j �x : A:K

A := x j �x : A:B j �x : A:B j AM

M := x j �x : A:M jMN

� Dependent Types: Types can depend on terms. In particular,

in �x : A:B, the variable x can occur in the type B. A ! B

denotes �x : A:B when x does not occur in B.

� Kinds can depend on terms also.

� Terms are similar to the �-terms of hohh except that in �x :

A:M , A can be a dependent type.

3

LF Contexts and Assertions

Syntax for Contexts (Signatures)

� := hi j �; x : K j �; x : A

LF Assertions

� ` K kind (K is a kind in �)

� ` A : K (A has kind K in �)

� `M : A (M has type A in �)

Valid Contexts

The empty context is valid and �; x : P is a valid context if � is a

valid context and either � ` P kind or � ` P : Type.

4

An LF Signature for Natural Deduction

A B

^-I

A ^B

(A)

B

�-I

A � B

[y=x]A

8-I

8xA

form : Type

i : Type

^ : form ! form ! form

8 : (i! form)! form

.
.
.

true : form ! Type

^-I : �A : form:�B : form:true(A)! true(B)! true(A ^B)

� -I : �A : form:�B : form:(true(A)! true(B))! true(A � B)

8-I : �A : i! form:(�x : i:true(Ax))! true(8A)

.
.
.

5



Translating Kind and Type Declarations

� Introducing New Base Types

form : Type

i : Type

kind form type.

kind i type.

� Introducing the Syntax of the Object Logic

^ : form ! form ! form

type and form -> form -> form.

� Dependent Type Constants as Predicates

true : form ! Type

type proof form -> o.

6

Inference Rules as Clauses I

^-I : �A : form:�B : form:true(A)! true(B)! true(A ^ B)

proof (A and B) :- proof A, proof B.

An LF term inhabiting the type true(A ^ B) will be a proof of the

formula A ^ B. If we use the above signature item in constructing

such a term, this term will have the form:

(^-I A B P

1

P

2

)

We can incorporate proof objects of this form into �Prolog speci�-

cations.

type proof nprf -> form -> o.

type and_i form -> form -> nprf -> nprf -> o.

proof (and_i A B P1 P2) (A and B) :-

proof P1 A, proof P2 B.

7

Inference Rules as Clauses II

� -I : �A : form:�B : form:(true(A)! true(B)) ! true(A � B)

type imp_i form -> form -> (nprf -> nprf) -> nprf.

proof (imp_i A B P) (A imp B) :-

pi pA\((proof pA A) => (proof (P pA) B)).

8-I : �A : i! form:(�x : i:true(Ax))! true(8A)

type forall_i (i -> form) -> (i -> nprf) -> nprf.

proof (forall_i A P) (forall A) :-

pi y\(proof (P y) (A y)).

Summary

� An LF signature item is translated to a type declaration and a

clause. The type declaration is a \at" version of the LF type,

while the clause replaces dependent types with predicates.

� This correspondence is formalized in [Felty&Miller, CADE'90].

� The translation is fairly direct, so the two are very close in

speci�cation strength.

� LF serves as a logical foundation for the logic programming

language Elf [Pfenning LICS'89].

